

D03PKF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

D03PKF integrates a system of linear or nonlinear, first-order, time-dependent partial differential equations (PDEs) in one space variable, with scope for coupled ordinary differential equations (ODEs). The spatial discretisation is performed using the Keller box scheme and the method of lines is employed to reduce the PDEs to a system of ODEs. The resulting system is solved using a Backward Differentiation Formula (BDF) method or a Theta method (switching between Newton's method and functional iteration).

2 Specification

```

SUBROUTINE D03PKF(NPDE, TS, TOUT, PDEDEF, BNDARY, U, NPTS, X,
1      NLEFT, NCODE, ODEDEF, NXI, XI, NEQN, RTOL, ATOL,
2      ITOL, NORM, LAOPT, ALGOPT, W, NW, IW, NIW,
3      ITASK, ITRACE, IND, IFAIL)
  INTEGER      NPDE, NPTS, NLEFT, NCODE, NXI, NEQN, ITOL, NW,
1      IW(NIW), NIW, ITASK, ITRACE, IND, IFAIL
  real         TS, TOUT, U(NEQN), X(NPTS), XI(*), RTOL(*),
1      ATOL(*), ALGOPT(30), W(NW)
  CHARACTER*1  NORM, LAOPT
  EXTERNAL    PDEDEF, BNDARY, ODEDEF

```

3 Description

D03PKF integrates the system of first-order PDEs and coupled ODEs

$$G_i(x, t, U, U_x, U_t, V, \dot{V}) = 0, \quad i = 1, 2, \dots, \text{NPDE}, \quad a \leq x \leq b, \quad t \geq t_0, \quad (1)$$

$$F_i(t, V, \dot{V}, \xi, U^*, U_x^*, U_t^*) = 0, \quad i = 1, 2, \dots, \text{NCODE}. \quad (2)$$

In the PDE part of the problem given by (1), the functions G_i must have the general form

$$G_i = \sum_{j=1}^{\text{NPDE}} P_{i,j} \frac{\partial U_j}{\partial t} + \sum_{j=1}^{\text{NCODE}} Q_{i,j} \dot{V}_j + R_i = 0, \quad i = 1, 2, \dots, \text{NPDE}, \quad (3)$$

where $P_{i,j}$, $Q_{i,j}$ and R_i depend on x, t, U, U_x and V .

The vector U is the set of PDE solution values

$$U(x, t) = [U_1(x, t), \dots, U_{\text{NPDE}}(x, t)]^T,$$

and the vector U_x is the partial derivative with respect to x . The vector V is the set of ODE solution values

$$V(t) = [V_1(t), \dots, V_{\text{NCODE}}(t)]^T,$$

and \dot{V} denotes its derivative with respect to time.

In the ODE part given by (2), ξ represents a vector of n_ξ spatial coupling points at which the ODEs are coupled to the PDEs. These points may or may not be equal to some of the PDE spatial mesh points. U^* , U_x^* and U_t^* are the functions U , U_x and U_t evaluated at these coupling points. Each F_i may only depend linearly on time derivatives. Hence equation (2) may be written more precisely as

$$F = A - B\dot{V} - CU_t^*, \quad (4)$$

where $F = [F_1, \dots, F_{\text{NCODE}}]^T$, A is a vector of length NCODE, B is an NCODE by NCODE matrix, C is an NCODE by $(n_\xi \times \text{NPDE})$ matrix. The entries in A , B and C may depend on t , ξ , U^* , U_x^* and V . In practice the user only needs to supply a vector of information to define the ODEs and not the matrices B and C . (See Section 5 for the specification of the user-supplied subroutine ODEDEF.)

The integration in time is from t_0 to t_{out} , over the space interval $a \leq x \leq b$, where $a = x_1$ and $b = x_{\text{NPTS}}$ are the leftmost and rightmost points of a user-defined mesh $x_1, x_2, \dots, x_{\text{NPTS}}$.

The PDE system which is defined by the functions G_i must be specified in the user-supplied subroutine PDEDEF.

The initial values of the functions $U(x, t)$ and $V(t)$ must be given at $t = t_0$.

For a first-order system of PDEs, only one boundary condition is required for each PDE component U_i . The NPDE boundary conditions are separated into NLEFT at the left-hand boundary $x = a$, and NRIGHT at the right-hand boundary $x = b$, such that NLEFT + NRIGHT = NPDE. The position of the boundary condition for each component should be chosen with care; the general rule is that if the characteristic direction of U_i at the left-hand boundary (say) points into the interior of the solution domain, then the boundary condition for U_i should be specified at the left-hand boundary. Incorrect positioning of boundary conditions generally results in initialisation or integration difficulties in the underlying time integration routines.

The boundary conditions have the form:

$$G_i^L(x, t, U, U_t, V, \dot{V}) = 0 \text{ at } x = a, \quad i = 1, 2, \dots, \text{NLEFT}, \quad (5)$$

at the left-hand boundary, and

$$G_i^R(x, t, U, U_t, V, \dot{V}) = 0 \text{ at } x = b, \quad i = 1, 2, \dots, \text{NRIGHT}, \quad (6)$$

at the right-hand boundary.

Note that the functions G_i^L and G_i^R must not depend on U_x , since spatial derivatives are not determined explicitly in the Keller box scheme. If the problem involves derivative (Neumann) boundary conditions then it is generally possible to restate such boundary conditions in terms of permissible variables. Also note that G_i^L and G_i^R must be linear with respect to time derivatives, so that the boundary conditions have the general form:

$$\sum_{j=1}^{\text{NPDE}} E_{i,j}^L \frac{\partial U_j}{\partial t} + \sum_{j=1}^{\text{NCODE}} H_{i,j}^L \dot{V}_j + S_i^L = 0, \quad i = 1, 2, \dots, \text{NLEFT}, \quad (7)$$

at the left-hand boundary, and

$$\sum_{j=1}^{\text{NPDE}} E_{i,j}^R \frac{\partial U_j}{\partial t} + \sum_{j=1}^{\text{NCODE}} H_{i,j}^R \dot{V}_j + S_i^R = 0, \quad i = 1, 2, \dots, \text{NRIGHT}, \quad (8)$$

at the right-hand boundary, where $E_{i,j}^L$, $E_{i,j}^R$, $H_{i,j}^L$, $H_{i,j}^R$, S_i^L and S_i^R depend on x, t, U and V only.

The boundary conditions must be specified in a subroutine BNDARY provided by the user.

The problem is subject to the following restrictions:

- (i) $P_{i,j}$, $Q_{i,j}$ and R_i must not depend on any time derivatives;
- (ii) $t_0 < t_{out}$, so that integration is in the forward direction;
- (iii) The evaluation of the function G_i is done approximately at the mid-points of the mesh $X(i)$, for $i = 1, 2, \dots, \text{NPTS}$, by calling the routine PDEDEF for each mid-point in turn. Any discontinuities in the function **must** therefore be at one or more of the mesh points $x_1, x_2, \dots, x_{\text{NPTS}}$;
- (iv) At least one of the functions $P_{i,j}$ must be non-zero so that there is a time derivative present in the PDE problem;

The algebraic-differential equation system which is defined by the functions F_i must be specified in the user-supplied subroutine ODEDEF. The user must also specify the coupling points ξ in the array XI.

The parabolic equations are approximated by a system of ODEs in time for the values of U_i at mesh points. In this method of lines approach the Keller box scheme [4] is applied to each PDE in the space variable only, resulting in a system of ODEs in time for the values of U_i at each mesh point. In total there are NPDE \times NPTS+NCODE ODEs in time direction. This system is then integrated forwards in time using a Backward Differentiation Formula (BDF) or a Theta method.

4 References

- [1] Berzins M (1990) Developments in the NAG Library software for parabolic equations *Scientific Software Systems* (ed J C Mason and M G Cox) Chapman and Hall 59–72
- [2] Berzins M, Dew P M and Furzeland R M (1989) Developing software for time-dependent problems using the method of lines and differential-algebraic integrators *Appl. Numer. Math.* **5** 375–397
- [3] Berzins M and Furzeland R M (1992) An adaptive theta method for the solution of stiff and nonstiff differential equations *Appl. Numer. Math.* **9** 1–19
- [4] Keller H B (1970) A new difference scheme for parabolic problems *Numerical Solutions of Partial Differential Equations* (ed J Bramble) **2** Academic Press 327–350
- [5] Pennington S V and Berzins M (1994) New NAG Library software for first-order partial differential equations *ACM Trans. Math. Softw.* **20** 63–99

5 Parameters

- 1: NPDE — INTEGER *Input*
On entry: the number of PDEs to be solved.
Constraint: NPDE \geq 1.
- 2: TS — *real* *Input/Output*
On entry: the initial value of the independent variable t .
Constraint: TS < TOUT.
On exit: the value of t corresponding to the solution in U. Normally TS = TOUT.
- 3: TOUT — *real* *Input*
On entry: the final value of t to which the integration is to be carried out.
- 4: PDEDEF — SUBROUTINE, supplied by the user. *External Procedure*
PDEDEF must evaluate the functions G_i which define the system of PDEs. PDEDEF is called approximately midway between each pair of mesh points in turn by D03PKF.

Its specification is:

```

SUBROUTINE PDEDEF(NPDE, T, X, U, UT, UX, NCODE, V, VDOT, RES, IRES)
  INTEGER          NPDE, NCODE, IRES
  real           T, X, U(NPDE), UT(NPDE), UX(NPDE), V(*),
  1               VDOT(*), RES(NPDE)

```

- 1: NPDE — INTEGER *Input*
On entry: the number of PDEs in the system.
- 2: T — *real* *Input*
On entry: the current value of the independent variable t .
- 3: X — *real* *Input*
On entry: the current value of the space variable x .
- 4: U(NPDE) — *real* array *Input*
On entry: U(i) contains the value of the component $U_i(x, t)$, for $i = 1, 2, \dots, \text{NPDE}$.
- 5: UT(NPDE) — *real* array *Input*
On entry: UT(i) contains the value of the component $\frac{\partial U_i(x, t)}{\partial t}$, for $i = 1, 2, \dots, \text{NPDE}$.

- 6:** UX(NPDE) — *real* array *Input*
On entry: UX(*i*) contains the value of the component $\frac{\partial U_i(x,t)}{\partial x}$, for $i = 1, 2, \dots, \text{NPDE}$.
- 7:** NCODE — INTEGER *Input*
On entry: the number of coupled ODEs in the system.
- 8:** V(*) — *real* array *Input*
On entry: V(*i*) contains the value of component $V_i(t)$, for $i = 1, 2, \dots, \text{NCODE}$.
- 9:** VDOT(*) — *real* array *Input*
On entry: VDOT(*i*) contains the value of component $\dot{V}_i(t)$, for $i = 1, 2, \dots, \text{NCODE}$.
- 10:** RES(NPDE) — *real* array *Output*
On exit: RES(*i*) must contain the *i*th component of G , for $i = 1, 2, \dots, \text{NPDE}$, where G is defined as

$$G_i = \sum_{j=1}^{\text{NPDE}} P_{i,j} \frac{\partial U_j}{\partial t} + \sum_{j=1}^{\text{NCODE}} Q_{i,j} \dot{V}_j, \quad (9)$$

i.e., only terms depending explicitly on time derivatives, or

$$G_i = \sum_{j=1}^{\text{NPDE}} P_{i,j} \frac{\partial U_j}{\partial t} + \sum_{j=1}^{\text{NCODE}} Q_{i,j} \dot{V}_j + R_i, \quad (10)$$

i.e., all terms in equation (3).

The definition of G is determined by the input value of IRES.

- 11:** IRES — INTEGER *Input/Output*
On entry: the form of G_i that must be returned in the array RES. If IRES = -1, then equation (9) above must be used. If IRES = 1, then equation (10) above must be used.
On exit: should usually remain unchanged. However, the user may set IRES to force the integration routine to take certain actions, as described below:

IRES = 2

indicates to the integrator that control should be passed back immediately to the calling (sub)program with the error indicator (IFAIL) set to 6.

IRES = 3

indicates to the integrator that the current time step should be abandoned and a smaller time step used instead. The user may wish to set IRES = 3 when a physically meaningless input or output value has been generated. If the user consecutively sets IRES = 3, then D03PKF returns to the calling (sub)program with the error indicator (IFAIL) set to 4.

PDEDEF must be declared as EXTERNAL in the (sub)program from which D03PKF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

- 5:** BNDARY — SUBROUTINE, supplied by the user. *External Procedure*
 BNDARY must evaluate the functions G_i^L and G_i^R which describe the boundary conditions, as given in (5) and (6).

Its specification is:

```

SUBROUTINE BNDARY(NPDE, T, IBND, NOBC, U, UT, NCODE, V, VDOT, RES,
1          IRES)
INTEGER    NPDE, IBND, NOBC, NCODE, IRES
real     T, U(NPDE), UT(NPDE), V(*), VDOT(*), RES(NOBC)

```

- 1:** NPDE — INTEGER *Input*
On entry: the number of PDEs in the system.
- 2:** T — **real** *Input*
On entry: the current value of the independent variable t .
- 3:** IBND — INTEGER *Input*
On entry: specifies which boundary conditions are to be evaluated. If IBND = 0, then BNDARY must compute the left-hand boundary condition at $x = a$. If IBND \neq 0, then BNDARY must compute the right-hand boundary condition at $x = b$.
- 4:** NOBC — INTEGER *Input*
On entry: NOBC specifies the number of boundary conditions at the boundary specified by IBND.
- 5:** U(NPDE) — **real** array *Input*
On entry: U(i) contains the value of the component $U_i(x, t)$ at the boundary specified by IBND, for $i = 1, 2, \dots, \text{NPDE}$.
- 6:** UT(NPDE) — **real** array *Input*
On entry: UT(i) contains the value of the component $\frac{\partial U_i(x, t)}{\partial t}$ at the boundary specified by IBND, for $i = 1, 2, \dots, \text{NPDE}$.
- 7:** NCODE — INTEGER *Input*
On entry: the number of coupled ODEs in the system.
- 8:** V(*) — **real** array *Input*
On entry: V(i) contains the value of component $V_i(t)$, for $i = 1, 2, \dots, \text{NCODE}$.
- 9:** VDOT(*) — **real** array *Input*
On entry: VDOT(i) contains the value of component $\dot{V}_i(t)$, for $i = 1, 2, \dots, \text{NCODE}$.
- Note.** VDOT(i), for $i = 1, 2, \dots, \text{NCODE}$, may only appear linearly as in (7) and (8).
- 10:** RES(NOBC) — **real** array *Output*
On exit: RES(i) must contain the i th component of G^L or G^R , depending on the value of IBND, for $i = 1, 2, \dots, \text{NOBC}$, where G^L is defined as

$$G_i^L = \sum_{j=1}^{\text{NPDE}} E_{i,j}^L \frac{\partial U_j}{\partial t} + \sum_{j=1}^{\text{NCODE}} H_{i,j}^L \dot{V}_j, \quad (11)$$

i.e., only terms depending explicitly on time derivatives, or

$$G_i^L = \sum_{j=1}^{\text{NPDE}} E_{i,j}^L \frac{\partial U_j}{\partial t} + \sum_{j=1}^{\text{NCODE}} H_{i,j}^L \dot{V}_j + S_i^L, \quad (12)$$

i.e., all terms in equation (7), and similarly for G_i^R .

The definitions of G^L and G^R are determined by the input value of IRES.

<p>11: IRES — INTEGER</p> <p><i>On entry:</i> the form of G_i^L (or G_i^R) that must be returned in the array RES. If IRES = -1, then equation (11) above must be used. If IRES = 1, then equation (12) above must be used.</p> <p><i>On exit:</i> should usually remain unchanged. However, the user may set IRES to force the integration routine to take certain actions, as described below:</p> <p>IRES = 2 indicates to the integrator that control should be passed back immediately to the calling (sub)program with the error indicator (IFAIL) set to 6.</p> <p>IRES = 3 indicates to the integrator that the current time step should be abandoned and a smaller time step used instead. The user may wish to set IRES = 3 when a physically meaningless input or output value has been generated. If the user consecutively sets IRES = 3, then D03PKF returns to the calling (sub)program with the error indicator (IFAIL) set to 4.</p>	<p><i>Input/Output</i></p>
--	----------------------------

BNDARY must be declared as EXTERNAL in the (sub)program from which D03PKF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

- 6:** U(NEQN) — *real* array *Input/Output*
- On entry:* the initial values of the dependent variables defined as follows:
- U(NPDE × (j-1)+i) contain $U_i(x_j, t_0)$, for $i = 1, 2, \dots, \text{NPDE}$; $j = 1, 2, \dots, \text{NPTS}$ and
- U(NPTS×NPDE+i) contain $V_i(t_0)$, for $i = 1, 2, \dots, \text{NCODE}$.
- On exit:* the computed solution $U_i(x_j, t)$, for $i = 1, 2, \dots, \text{NPDE}$; $j = 1, 2, \dots, \text{NPTS}$, and $V_k(t)$, for $k = 1, 2, \dots, \text{NCODE}$, evaluated at $t = \text{TS}$.
- 7:** NPTS — INTEGER *Input*
- On entry:* the number of mesh points in the interval $[a, b]$.
- Constraint:* $\text{NPTS} \geq 3$.
- 8:** X(NPTS) — *real* array *Input*
- On entry:* the mesh points in the space direction. X(1) must specify the left-hand boundary, a , and X(NPTS) must specify the right-hand boundary, b .
- Constraint:* $X(1) < X(2) < \dots < X(\text{NPTS})$.
- 9:** NLEFT — INTEGER *Input*
- On entry:* the number of boundary conditions at the left-hand mesh point X(1).
- Constraint:* $0 \leq \text{NLEFT} \leq \text{NPDE}$.
- 10:** NCODE — INTEGER *Input*
- On entry:* the number of coupled ODE components.
- Constraint:* $\text{NCODE} \geq 0$.
- 11:** ODEDEF — SUBROUTINE, supplied by the user. *External Procedure*
- ODEDEF must evaluate the functions F , which define the system of ODEs, as given in (4). If the user wishes to compute the solution of a system of PDEs only (i.e., $\text{NCODE} = 0$), ODEDEF must be the dummy routine D03PEK. (D03PEK is included in the NAG Fortran Library; however, its name may be implementation-dependent: see the Users' Note for your implementation for details.)

Its specification is:

```

SUBROUTINE ODEDEF(NPDE, T, NCODE, V, VDOT, NXI, XI, UCP, UCPX,
1          UCPT, F, IRES)
INTEGER    NPDE, NCODE, NXI, IRES
real     T, V(*), VDOT(*), XI(*), UCP(NPDE,*),
1          UCPX(NPDE,*), UCPT(NPDE,*), F(*)

```

- 1:** NPDE — INTEGER *Input*
On entry: the number of PDEs in the system.
- 2:** T — *real* *Input*
On entry: the current value of the independent variable t .
- 3:** NCODE — INTEGER *Input*
On entry: the number of coupled ODEs in the system.
- 4:** V(*) — *real* array *Input*
On entry: V(i) contains the value of component $V_i(t)$, for $i = 1, 2, \dots, \text{NCODE}$.
- 5:** VDOT(*) — *real* array *Input*
On entry: VDOT(i) contains the value of component $\dot{V}_i(t)$, for $i = 1, 2, \dots, \text{NCODE}$.
- 6:** NXI — INTEGER *Input*
On entry: the number of ODE/PDE coupling points.
- 7:** XI(*) — *real* array *Input*
On entry: XI(i) contains the ODE/PDE coupling point ξ_i , for $i = 1, 2, \dots, \text{NXI}$.
- 8:** UCP(NPDE,*) — *real* array *Input*
On entry: UCP(i, j) contains the value of $U_i(x, t)$ at the coupling point $x = \xi_j$, for $i = 1, 2, \dots, \text{NPDE}$; $j = 1, 2, \dots, \text{NXI}$.
- 9:** UCPX(NPDE,*) — *real* array *Input*
On entry: UCPX(i, j) contains the value of $\frac{\partial U_i(x, t)}{\partial x}$ at the coupling point $x = \xi_j$, for $i = 1, 2, \dots, \text{NPDE}$; $j = 1, 2, \dots, \text{NXI}$.
- 10:** UCPT(NPDE,*) — *real* array *Input*
On entry: UCPT(i, j) contains the value of $\frac{\partial U_i(x, t)}{\partial t}$ at the coupling point $x = \xi_j$, for $i = 1, 2, \dots, \text{NPDE}$; $j = 1, 2, \dots, \text{NXI}$.
- 11:** F(*) — *real* array *Output*
On exit: F(i) must contain the i th component of F , for $i = 1, 2, \dots, \text{NCODE}$, where F is defined as

$$F = -B\dot{V} - CU_t^*, \quad (13)$$

i.e., only terms depending explicitly on time derivatives, or

$$F = A - B\dot{V} - CU_t^*, \quad (14)$$

i.e., all terms in equation (4). The definition of F is determined by the input value of IRES.

- 12:** IRES — INTEGER *Input/Output*
On entry: the form of F that must be returned in the array F. If IRES = -1, then equation (13) above must be used. If IRES = 1, then equation (14) above must be used.
On exit: should usually remain unchanged. However, the user may reset IRES to force the integration routine to take certain actions, as described below:

IRES = 2

indicates to the integrator that control should be passed back immediately to the calling (sub)program with the error indicator (IFAIL) set to 6.

IRES = 3

indicates to the integrator that the current time step should be abandoned and a smaller time step used instead. The user may wish to set IRES = 3 when a physically meaningless input or output value has been generated. If the user consecutively sets IRES = 3, then D03PKF returns to the calling (sub)program with the error indicator (IFAIL) set to 4.

ODEDEF must be declared as EXTERNAL in the (sub)program from which D03PKF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

12: NXI — INTEGER *Input*

Constraints:

NXI = 0 for NCODE = 0.

NXI ≥ 0 for NCODE > 0.

13: XI(*) — *real* array *Input*

Note: the dimension of the array XI must be at least max(1,NXI).

On entry: XI(*i*), *i* = 1, 2, ..., NXI, must be set to the ODE/PDE coupling points, ξ_i .

Constraint: X(1) ≤ XI(1) < XI(2) < ... < XI(NXI) ≤ X(NPTS).

14: NEQN — INTEGER *Input*

On entry: the number of ODEs in the time direction.

Constraint: NEQN = NPDE × NPTS + NCODE.

15: RTOL(*) — *real* array *Input*

Note: the dimension of the array RTOL must be at least 1 if ITOL = 1 or 2 and at least NEQN if ITOL = 3 or 4.

On entry: the relative local error tolerance.

Constraint: RTOL(*i*) ≥ 0 for all relevant *i*.

16: ATOL(*) — *real* array *Input*

Note: the dimension of the array ATOL must be at least 1 if ITOL = 1 or 3 and at least NEQN if ITOL = 2 or 4.

On entry: the absolute local error tolerance.

Constraints: ATOL(*i*) ≥ 0 for all relevant *i*.

Corresponding elements of ATOL and RTOL should not both be 0.0.

17: ITOL — INTEGER *Input*

On entry: a value to indicate the form of the local error test. ITOL indicates to D03PKF whether to interpret either or both of RTOL or ATOL as a vector or scalar. The error test to be satisfied is $\|e_i/w_i\| < 1.0$, where w_i is defined as follows:

ITOL	RTOL	ATOL	w_i
1	scalar	scalar	RTOL(1) × U(<i>i</i>) + ATOL(1)
2	scalar	vector	RTOL(1) × U(<i>i</i>) + ATOL(<i>i</i>)
3	vector	scalar	RTOL(<i>i</i>) × U(<i>i</i>) + ATOL(1)
4	vector	vector	RTOL(<i>i</i>) × U(<i>i</i>) + ATOL(<i>i</i>)

In the above, e_i denotes the estimated local error for the i th component of the coupled PDE/ODE system in time, $U(i)$, for $i = 1, 2, \dots, \text{NEQN}$.

The choice of norm used is defined by the parameter `NORM`, see below.

Constraint: $1 \leq \text{ITOL} \leq 4$.

18: `NORM` — CHARACTER*1

Input

On entry: the type of norm to be used. Two options are available:

- 'M' – maximum norm.
- 'A' – averaged L_2 norm.

If U_{norm} denotes the norm of the vector U of length `NEQN`, then for the averaged L_2 norm

$$U_{\text{norm}} = \sqrt{\frac{1}{\text{NEQN}} \sum_{i=1}^{\text{NEQN}} (U(i)/w_i)^2},$$

while for the maximum norm

$$U_{\text{norm}} = \max_i |U(i)/w_i|.$$

See the description of the `ITOL` parameter for the formulation of the weight vector w .

Constraint: `NORM` = 'M' or 'A'.

19: `LAOPT` — CHARACTER*1

Input

On entry: the type of matrix algebra required. The possible choices are:

- 'F' – full matrix routines to be used;
- 'B' – banded matrix routines to be used;
- 'S' – sparse matrix routines to be used.

Constraint: `LAOPT` = 'F', 'B' or 'S'.

Note. The user is recommended to use the banded option when no coupled ODEs are present (i.e., `NCODE` = 0).

20: `ALGOPT(30)` — *real* array

Input

On entry: `ALGOPT` may be set to control various options available in the integrator. If the user wishes to employ all the default options, then `ALGOPT(1)` should be set to 0.0. Default values will also be used for any other elements of `ALGOPT` set to zero. The permissible values, default values, and meanings are as follows:

`ALGOPT(1)` selects the ODE integration method to be used. If `ALGOPT(1) = 1.0`, a BDF method is used and if `ALGOPT(1) = 2.0`, a Theta method is used.

The default value is `ALGOPT(1) = 1.0`.

If `ALGOPT(1) = 2.0`, then `ALGOPT(i)`, for $i = 2, 3, 4$ are not used.

`ALGOPT(2)` specifies the maximum order of the BDF integration formula to be used. `ALGOPT(2)` may be 1.0, 2.0, 3.0, 4.0 or 5.0.

The default value is `ALGOPT(2) = 5.0`.

`ALGOPT(3)` specifies what method is to be used to solve the system of nonlinear equations arising on each step of the BDF method. If `ALGOPT(3) = 1.0` a modified Newton iteration is used and if `ALGOPT(3) = 2.0` a functional iteration method is used. If functional iteration is selected and the integrator encounters difficulty, then there is an automatic switch to the modified Newton iteration.

The default value is `ALGOPT(3) = 1.0`.

ALGOPT(4) specifies whether or not the Petzold error test is to be employed. The Petzold error test results in extra overhead but is more suitable when algebraic equations are present, such as $P_{i,j} = 0.0$, for $j = 1, 2, \dots, \text{NPDE}$ for some i or when there is no $\dot{V}_i(t)$ dependence in the coupled ODE system. If ALGOPT(4) = 1.0, then the Petzold test is used. If ALGOPT(4) = 2.0, then the Petzold test is not used.

The default value is ALGOPT(4) = 1.0.

If ALGOPT(1) = 1.0, then ALGOPT(i), for $i = 5, 6, 7$ are not used.

ALGOPT(5), specifies the value of Theta to be used in the Theta integration method.

$0.51 \leq \text{ALGOPT}(5) \leq 0.99$.

The default value is ALGOPT(5) = 0.55.

ALGOPT(6) specifies what method is to be used to solve the system of nonlinear equations arising on each step of the Theta method. If ALGOPT(6) = 1.0, a modified Newton iteration is used and if ALGOPT(6) = 2.0, a functional iteration method is used.

The default value is ALGOPT(6) = 1.0.

ALGOPT(7) specifies whether or not the integrator is allowed to switch automatically between modified Newton and functional iteration methods in order to be more efficient. If ALGOPT(7) = 1.0, then switching is allowed and if ALGOPT(7) = 2.0, then switching is not allowed.

The default value is ALGOPT(7) = 1.0.

ALGOPT(11) specifies a point in the time direction, t_{crit} , beyond which integration must not be attempted. The use of t_{crit} is described under the parameter ITASK. If ALGOPT(1) \neq 0.0, a value of 0.0 for ALGOPT(11), say, should be specified even if ITASK subsequently specifies that t_{crit} will not be used.

ALGOPT(12) specifies the minimum absolute step size to be allowed in the time integration. If this option is not required, ALGOPT(12) should be set to 0.0.

ALGOPT(13) specifies the maximum absolute step size to be allowed in the time integration. If this option is not required, ALGOPT(13) should be set to 0.0.

ALGOPT(14) specifies the initial step size to be attempted by the integrator. If ALGOPT(14) = 0.0, then the initial step size is calculated internally.

ALGOPT(15) specifies the maximum number of steps to be attempted by the integrator in any one call. If ALGOPT(15) = 0.0, then no limit is imposed.

ALGOPT(23) specifies what method is to be used to solve the nonlinear equations at the initial point to initialise the values of U , U_t , V and \dot{V} . If ALGOPT(23) = 1.0, a modified Newton iteration is used and if ALGOPT(23) = 2.0, functional iteration is used.

The default value is ALGOPT(23) = 1.0.

ALGOPT(29) and ALGOPT(30) are used only for the sparse matrix algebra option, i.e., LAOPT = 'S'.

ALGOPT(29) governs the choice of pivots during the decomposition of the first Jacobian matrix. It should lie in the range $0.0 < \text{ALGOPT}(29) < 1.0$, with smaller values biasing the algorithm towards maintaining sparsity at the expense of numerical stability. If ALGOPT(29) lies outside this range then the default value is used. If the routines regard the Jacobian matrix as numerically singular then increasing ALGOPT(29) towards 1.0 may help, but at the cost of increased fill-in.

The default value is ALGOPT(29) = 0.1.

ALGOPT(30) is used as a relative pivot threshold during subsequent Jacobian decompositions (see ALGOPT(29)) below which an internal error is invoked. ALGOPT(30) must be greater than zero, otherwise the default value is used. If ALGOPT(30) is greater than 1.0 no check is made on the pivot size, and this may be a necessary option if the Jacobian is found to be numerically singular (see ALGOPT(29)).

The default value is $\text{ALGOPT}(30) = 0.0001$.

- 21:** W(NW) — *real* array *Workspace*
22: NW — INTEGER *Input*

On entry: the dimension of the array W as declared in the (sub)program from which D03PKF is called. Its size depends on the type of matrix algebra selected:

LAOPT = 'F',
 $NW \geq \text{NEQN} \times \text{NEQN} + \text{NEQN} + \text{NWKRES} + \text{LENODE}$,

LAOPT = 'B',
 $NW \geq (2 \times \text{ML} + \text{MU} + 2) \times \text{NEQN} + \text{NWKRES} + \text{LENODE}$,

LAOPT = 'S',
 $NW \geq 4 \times \text{NEQN} + 11 \times \text{NEQN}/2 + 1 + \text{NWKRES} + \text{LENODE}$,

where ML and MU are the lower and upper half bandwidths, given by $\text{ML} = \text{NPDE} + \text{NLEFT} - 1$, $\text{MU} = 2 \times \text{NPDE} - \text{NLEFT} - 1$ for problems involving PDEs only, and $\text{ML} = \text{MU} = \text{NEQN} - 1$, for coupled PDE/ODE problems.

$\text{NWKRES} = \text{NPDE} \times (6 \times \text{NXI} + 3 \times \text{NPDE} + \text{NPTS} + 15) + \text{NXI} + \text{NCODE} + 7 \times \text{NPTS} + 2$

when $\text{NCODE} > 0$, and $\text{NXI} > 0$.

$\text{NWKRES} = \text{NPDE} \times (3 \times \text{NPDE} + \text{NPTS} + 21) + \text{NCODE} + 7 \times \text{NPTS} + 3$

when $\text{NCODE} > 0$, and $\text{NXI} = 0$.

$\text{NWKRES} = \text{NPDE} \times (3 \times \text{NPDE} + \text{NPTS} + 21) + 7 \times \text{NPTS} + 4$

when $\text{NCODE} = 0$.

$\text{LENODE} = (6 + \text{int}(\text{ALGOPT}(2))) \times \text{NEQN} + 50$, when the BDF method is used and,

$\text{LENODE} = 9 \times \text{NEQN} + 50$, when the Theta method is used.

Note. When using the sparse option, the value of NW may be too small when supplied to the integrator. An estimate of the minimum size of NW is printed on the current error message unit if $\text{ITRACE} > 0$ and the routine returns with $\text{IFAIL} = 15$.

- 23:** IW(NIW) — INTEGER array *Output*

On exit: the following components of the array IW concern the efficiency of the integration.

IW(1) contains the number of steps taken in time.

IW(2) contains the number of residual evaluations of the resulting ODE system used. One such evaluation involves evaluating the PDE functions at all the mesh points, as well as one evaluation of the functions in the boundary conditions.

IW(3) contains the number of Jacobian evaluations performed by the time integrator.

IW(4) contains the order of the ODE method last used in the time integration.

IW(5) contains the number of Newton iterations performed by the time integrator. Each iteration involves residual evaluation of the resulting ODE system followed by a back-substitution using the LU decomposition of the Jacobian matrix.

The rest of the array is used as workspace.

24: NIW — INTEGER *Input*

On entry: the dimension of the array IW. Its size depends on the type of matrix algebra selected:

LAOPT = 'F',
NIW \geq 24,

LAOPT = 'B',
NIW \geq NEQN+24,

LAOPT = 'S',
NIW \geq 25 \times NEQN+24.

Note. When using the sparse option, the value of NIW may be too small when supplied to the integrator. An estimate of the minimum size of NIW is printed on the current error message unit if ITRACE > 0 and the routine returns with IFAIL = 15.

25: ITASK — INTEGER *Input*

On entry: the task to be performed by the ODE integrator. The permitted values of ITASK and their meanings are detailed below:

ITASK = 1
normal computation of output values U at $t = \text{TOUT}$ (by overshooting and interpolating).

ITASK = 2
take one step in the time direction and return.

ITASK = 3
stop at first internal integration point at or beyond $t = \text{TOUT}$.

ITASK = 4
normal computation of output values U at $t = \text{TOUT}$ but without overshooting $t = t_{\text{crit}}$, where t_{crit} is described under the parameter ALGOPT.

ITASK = 5
take one step in the time direction and return, without passing t_{crit} , where t_{crit} is described under the parameter ALGOPT.

Constraint: $1 \leq \text{ITASK} \leq 5$.

26: ITRACE — INTEGER *Input*

On entry: the level of trace information required from D03PKF and the underlying ODE solver as follows:

If ITRACE ≤ -1 , no output is generated.

If ITRACE = 0, only warning messages from the PDE solver are printed on the current error message unit (see X04AAF).

If ITRACE = 1, then output from the underlying ODE solver is printed on the current advisory message unit (see X04ABF). This output contains details of Jacobian entries, the nonlinear iteration and the time integration during the computation of the ODE system.

If ITRACE = 2, then the output from the underlying ODE solver is similar to that produced when ITRACE = 1, except that the advisory messages are given in greater detail.

If ITRACE ≥ 3 , then the output from the underlying ODE solver is similar to that produced when ITRACE = 2, except that the advisory messages are given in greater detail.

Users are advised to set ITRACE = 0, unless they are experienced with the subchapter D02M–N of the NAG Fortran Library.

27: IND — INTEGER*Input/Output*

On entry: IND must be set to 0 or 1.

IND = 0

starts or restarts the integration in time.

IND = 1

continues the integration after an earlier exit from the routine. In this case, only the parameters TOUT and IFAIL should be reset between calls to D03PKF.

Constraint: $0 \leq \text{IND} \leq 1$.

On exit: IND = 1.

28: IFAIL — INTEGER*Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

On entry, (TOUT-TS) is too small,

or ITASK \neq 1, 2, 3, 4 or 5,

or at least one of the coupling points defined in array XI is outside the interval $[X(1), X(NPTS)]$,

or NPTS < 3,

or NPDE < 1,

or NLEFT not in range 0 to NPDE,

or NORM \neq 'A' or 'M',

or LAOPT \neq 'F', 'B' or 'S',

or ITOL \neq 1, 2, 3 or 4,

or IND \neq 0 or 1,

or incorrectly defined user mesh, i.e., $X(i) \geq X(i+1)$ for some $i = 1, 2, \dots, NPTS-1$,

or NW or NIW are too small,

or NCODE and NXI are incorrectly defined,

or IND = 1 on initial entry to D03PKF,

or an element of RTOL or ATOL < 0.0,

or corresponding elements of ATOL and RTOL are both 0.0,

or NEQN \neq NPDE \times NPTS+NCODE.

IFAIL = 2

The underlying ODE solver cannot make any further progress, with the values of ATOL and RTOL, across the integration range from the current point $t = \text{TS}$. The components of U contain the computed values at the current point $t = \text{TS}$.

IFAIL = 3

In the underlying ODE solver, there were repeated error test failures on an attempted step, before completing the requested task, but the integration was successful as far as $t = \text{TS}$. The problem may have a singularity, or the error requirement may be inappropriate. Incorrect positioning of boundary conditions may also result in this error.

IFAIL = 4

In setting up the ODE system, the internal initialisation routine was unable to initialise the derivative of the ODE system. This could be due to the fact that IRES was repeatedly set to 3 in one of the user-supplied subroutines PDEDEF, BNDARY or ODEDEF, when the residual in the underlying ODE solver was being evaluated. Incorrect positioning of boundary conditions may also result in this error.

IFAIL = 5

In solving the ODE system, a singular Jacobian has been encountered. The user should check their problem formulation.

IFAIL = 6

When evaluating the residual in solving the ODE system, IRES was set to 2 in one of the user-supplied subroutines PDEDEF, BNDARY or ODEDEF. Integration was successful as far as $t = TS$.

IFAIL = 7

The values of ATOL and RTOL are so small that the routine is unable to start the integration in time.

IFAIL = 8

In one of the user-supplied routines, PDEDEF, BNDARY or ODEDEF, IRES was set to an invalid value.

IFAIL = 9

A serious error has occurred in an internal call to D02NNF. Check problem specification and all parameters and array dimensions. Setting ITRACE = 1 may provide more information. If the problem persists, contact NAG.

IFAIL = 10

The required task has been completed, but it is estimated that a small change in ATOL and RTOL is unlikely to produce any change in the computed solution. (Only applies when the user is not operating in one step mode, that is when ITASK \neq 2 or 5.)

IFAIL = 11

An error occurred during Jacobian formulation of the ODE system (a more detailed error description may be directed to the current advisory message unit). If using the sparse matrix algebra option, the values of ALGOPT(29) and ALGOPT(30) may be inappropriate.

IFAIL = 12

In solving the ODE system, the maximum number of steps specified in ALGOPT(15) has been taken.

IFAIL = 13

Some error weights w_i became zero during the time integration (see description of ITOL). Pure relative error control ($ATOL(i) = 0.0$) was requested on a variable (the i th) which has become zero. The integration was successful as far as $t = TS$.

IFAIL = 14

Not applicable.

IFAIL = 15

When using the sparse option, the value of NIW or NW was insufficient (more detailed information may be directed to the current error message unit).

7 Accuracy

The routine controls the accuracy of the integration in the time direction but not the accuracy of the approximation in space. The spatial accuracy depends on both the number of mesh points and on their distribution in space. In the time integration only the local error over a single step is controlled and so the accuracy over a number of steps cannot be guaranteed. The user should therefore test the effect of varying the accuracy parameters, ATOL and RTOL.

8 Further Comments

The Keller box scheme can be used to solve higher-order problems which have been reduced to first-order by the introduction of new variables (see the example in Section 9 below). In general, a second-order problem can be solved with slightly greater accuracy using the Keller box scheme instead of a finite-difference scheme (D03PCF/D03PHF for example), but at the expense of increased CPU time due to the larger number of function evaluations required.

It should be noted that the Keller box scheme, in common with other central-difference schemes, may be unsuitable for some hyperbolic first-order problems such as the apparently simple linear advection equation $U_t + aU_x = 0$, where a is a constant, resulting in spurious oscillations due to the lack of dissipation. This type of problem requires a discretisation scheme with upwind weighting (D03PLF for example), or the addition of a second-order artificial dissipation term.

The time taken by the routine depends on the complexity of the system and on the accuracy requested. For a given system and a fixed accuracy it is approximately proportional to NEQN.

9 Example

This problem provides a simple coupled system of two PDEs and one ODE.

$$(V_1)^2 \frac{\partial U_1}{\partial t} - xV_1 \dot{V}_1 U_2 - \frac{\partial U_2}{\partial x} = 0,$$

$$U_2 - \frac{\partial U_1}{\partial x} = 0,$$

$$\dot{V}_1 - V_1 U_1 - U_2 - 1 - t = 0,$$

for $t \in [10^{-4}, 0.1 \times 2^i]$, for $i = 1, 2, \dots, 5$, $x \in [0, 1]$. The left boundary condition at $x = 0$ is

$$U_2 = -V_1 \exp t,$$

and the right boundary condition at $x = 1$ is

$$U_2 = -V_1 \dot{V}_1.$$

The initial conditions at $t = 10^{-4}$ are defined by the exact solution:

$$V_1 = t, \quad U_1(x, t) = \exp\{t(1-x)\} - 1.0 \quad \text{and} \quad U_2(x, t) = -t \exp\{t(1-x)\}, \quad x \in [0, 1],$$

and the coupling point is at $\xi_1 = 1.0$.

This problem is exactly the same as the D03PHF example problem, but reduced to first-order by the introduction of a second PDE variable (as mentioned in Section 8).

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*   D03PKF Example Program Text
*   Mark 16 Release. NAG Copyright 1993.
*   .. Parameters ..
INTEGER          NOUT
PARAMETER       (NOUT=6)
INTEGER         NPDE, NPTS, NCODE, NXI, NLEFT, NEQN, NIW, NWKRES,
+              LENODE, NW
PARAMETER       (NPDE=2, NPTS=21, NCODE=1, NXI=1, NLEFT=1,
+              NEQN=NPDE*NPTS+NCODE, NIW=24,
+              NWKRES=NPDE*(NPTS+6*NXI+3*NPDE+15)
+              +NCODE+NXI+7*NPTS+2, LENODE=11*NEQN+50,
+              NW=NEQN*NEQN+NEQN+NWKRES+LENODE)
*   .. Scalars in Common ..
  real          TS
*   .. Local Scalars ..
  real          TOUT
INTEGER         I, IFAIL, IND, IT, ITASK, ITOL, ITRACE
LOGICAL         THETA
CHARACTER       LAOPT, NORM
*   .. Local Arrays ..
  real          ALGOPT(30), ATOL(1), EXY(NEQN), RTOL(1), U(NEQN),
+              W(NW), X(NPTS), XI(1)
INTEGER         IW(NIW)
*   .. External Subroutines ..
EXTERNAL        BNDARY, D03PKF, EXACT, ODEDEF, PDEDEF, UVINIT
*   .. Common blocks ..
COMMON          /TAXIS/TS
*   .. Executable Statements ..
WRITE (NOUT,*) 'D03PKF Example Program Results'
ITRACE = 0
ITOL = 1
ATOL(1) = 0.1e-3
RTOL(1) = ATOL(1)
WRITE (NOUT,99997) ATOL, NPTS
*
*   Set spatial-mesh points
*
DO 20 I = 1, NPTS
    X(I) = (I-1.0e0)/(NPTS-1.0e0)
20 CONTINUE
*
    XI(1) = 1.0e0
    NORM = 'A'
    LAOPT = 'F'
    IND = 0
    ITASK = 1
*
*   Set THETA to .TRUE. if the Theta integrator is required
*
    THETA = .FALSE.
DO 40 I = 1, 30
    ALGOPT(I) = 0.0e0
40 CONTINUE
    IF (THETA) THEN

```



```

        ALGOPT(1) = 2.0e0
    ELSE
        ALGOPT(1) = 0.0e0
    END IF
    ALGOPT(1) = 1.0e0
    ALGOPT(13) = 0.5e-2
*
*   Loop over output value of t
*
    TS = 1.0e-4
    TOUT = 0.0e0
    WRITE (NOUT,99999) X(1), X(5), X(9), X(13), X(21)
*
    CALL UVINIT(NPDE,NPTS,X,U,NCODE,NEQN)
*
    DO 60 IT = 1, 5
        TOUT = 0.1e0*(2**IT)
        IFAIL = -1
*
        CALL D03PKF(NPDE,TS,TOUT,PDEDEF,BNDARY,U,NPTS,X,NLEFT,NCODE,
+                ODEDEF,NXI,XI,NEQN,RTOL,ATOL,ITOL,NORM,LAOPT,
+                ALGOPT,W,NW,IW,NIW,ITASK,ITRACE,IND,IFAIL)
*
        Check against the exact solution
*
        CALL EXACT(TOUT,NEQN,NPTS,X,EXY)
*
        WRITE (NOUT,99998) TS
        WRITE (NOUT,99995) U(1), U(9), U(17), U(25), U(41), U(43)
        WRITE (NOUT,99994) EXY(1), EXY(9), EXY(17), EXY(25), EXY(41),
+        TS
    60 CONTINUE
        WRITE (NOUT,99996) IW(1), IW(2), IW(3), IW(5)
        STOP
*
99999 FORMAT (' X          ',5F9.3,/)
99998 FORMAT (' T = ',F6.3)
99997 FORMAT (//' Accuracy requirement =',e10.3,' Number of points = ',
+           I3,/)
99996 FORMAT (' Number of integration steps in time = ',I6,/' Number o',
+           'f function evaluations = ',I6,/' Number of Jacobian eval',
+           'uations =',I6,/' Number of iterations = ',I6,/)
99995 FORMAT (1X,'App. sol. ',F7.3,4F9.3,' ODE sol. =',F8.3)
99994 FORMAT (1X,'Exact sol. ',F7.3,4F9.3,' ODE sol. =',F8.3,/)
    END
*
    SUBROUTINE UVINIT(NPDE,NPTS,X,U,NCODE,NEQN)
*
    Routine for PDE initial values
*
    .. Scalar Arguments ..
    INTEGER          NCODE, NEQN, NPDE, NPTS
*
    .. Array Arguments ..
    real             U(NEQN), X(NPTS)
*
    .. Scalars in Common ..
    real             TS
*
    .. Local Scalars ..
    INTEGER          I, K
*
    .. Intrinsic Functions ..
    INTRINSIC       EXP

```

```

*      .. Common blocks ..
COMMON          /TAXIS/TS
*      .. Executable Statements ..
K = 1
DO 20 I = 1, NPTS
    U(K) = EXP(TS*(1.0e0-X(I))) - 1.0e0
    U(K+1) = -TS*EXP(TS*(1.0e0-X(I)))
    K = K + 2
20 CONTINUE
U(NEQN) = TS
RETURN
END

*
SUBROUTINE ODEDEF(NPDE,T,NCODE,V,VDOT,NXI,XI,UCP,UCPX,UCPT,F,IRES)
*      .. Scalar Arguments ..
real          T
INTEGER        IRES, NCODE, NPDE, NXI
*      .. Array Arguments ..
real          F(*), UCP(NPDE,*), UCPT(NPDE,*), UCPX(NPDE,*),
+              V(*), VDOT(*), XI(*)
*      .. Executable Statements ..
IF (IRES.EQ.-1) THEN
    F(1) = VDOT(1)
ELSE
    F(1) = VDOT(1) - V(1)*UCP(1,1) - UCP(2,1) - 1.0e0 - T
END IF
RETURN
END

*
SUBROUTINE PDEDEF(NPDE,T,X,U,UDOT,UX,NCODE,V,VDOT,RES,IRES)
*      .. Scalar Arguments ..
real          T, X
INTEGER        IRES, NCODE, NPDE
*      .. Array Arguments ..
real          RES(NPDE), U(NPDE), UDOT(NPDE), UX(NPDE), V(*),
+              VDOT(*)
*      .. Executable Statements ..
IF (IRES.EQ.-1) THEN
    RES(1) = V(1)*V(1)*UDOT(1) - X*U(2)*V(1)*VDOT(1)
    RES(2) = 0.0e0
ELSE
    RES(1) = V(1)*V(1)*UDOT(1) - X*U(2)*V(1)*VDOT(1) - UX(2)
    RES(2) = U(2) - UX(1)
END IF
RETURN
END

*
SUBROUTINE BNDARY(NPDE,T,IBND,NOBC,U,UDOT,NCODE,V,VDOT,RES,IRES)
*      .. Scalar Arguments ..
real          T
INTEGER        IBND, IRES, NCODE, NOBC, NPDE
*      .. Array Arguments ..
real          RES(NOBC), U(NPDE), UDOT(NPDE), V(*), VDOT(*)
*      .. Intrinsic Functions ..
INTRINSIC      EXP
*      .. Executable Statements ..
IF (IBND.EQ.0) THEN
    IF (IRES.EQ.-1) THEN

```

```

        RES(1) = 0.0e0
    ELSE
        RES(1) = U(2) + V(1)*EXP(T)
    END IF
ELSE
    IF (IRES.EQ.-1) THEN
        RES(1) = V(1)*VDOT(1)
    ELSE
        RES(1) = U(2) + V(1)*VDOT(1)
    END IF
END IF
RETURN
END

*
SUBROUTINE EXACT(TIME,NEQN,NPTS,X,U)
* Exact solution (for comparison purposes)
* .. Scalar Arguments ..
  real          TIME
  INTEGER       NEQN, NPTS
* .. Array Arguments ..
  real          U(NEQN), X(NPTS)
* .. Local Scalars ..
  INTEGER       I, K
* .. Intrinsic Functions ..
  INTRINSIC    EXP
* .. Executable Statements ..
  K = 1
  DO 20 I = 1, NPTS
    U(K) = EXP(TIME*(1.0e0-X(I))) - 1.0e0
    K = K + 2
  20 CONTINUE
  RETURN
  END

```

9.2 Program Data

None.

9.3 Program Results

D03PKF Example Program Results

Accuracy requirement = 0.100E-03 Number of points = 21

X	0.000	0.200	0.400	0.600	1.000		
T = 0.200							
App. sol.	0.222	0.174	0.128	0.084	0.000	ODE sol. =	0.200
Exact sol.	0.221	0.174	0.127	0.083	0.000	ODE sol. =	0.200
T = 0.400							
App. sol.	0.492	0.377	0.271	0.174	0.000	ODE sol. =	0.400
Exact sol.	0.492	0.377	0.271	0.174	0.000	ODE sol. =	0.400
T = 0.800							
App. sol.	1.226	0.896	0.616	0.377	0.000	ODE sol. =	0.800
Exact sol.	1.226	0.896	0.616	0.377	0.000	ODE sol. =	0.800

T = 1.600
App. sol. 3.952 2.595 1.610 0.895 -0.001 ODE sol. = 1.600
Exact sol. 3.953 2.597 1.612 0.896 0.000 ODE sol. = 1.600

T = 3.200
App. sol. 23.522 11.918 5.807 2.588 -0.004 ODE sol. = 3.197
Exact sol. 23.533 11.936 5.821 2.597 0.000 ODE sol. = 3.200

Number of integration steps in time = 642
Number of function evaluations = 3022
Number of Jacobian evaluations = 39
Number of iterations = 1328
